；使用左右方向键控制篮子移动方向
#窗 = @生成图形、《窗口 #名=“#窗”，#标题=“接鸡蛋”，#宽=360，#高=695 【
    《图片 #宽=350，#高=655，#源=“./res/img/背景.jpg”，#适应=“覆盖”》
    《图片 #宽=75，#高=83，#源=“./res/img/鸡.jpg”，#适应=“覆盖”，#X=10，#Y=20》
    《图片 #宽=75，#高=83，#源=“./res/img/鸡.jpg”，#适应=“覆盖”，#X=95，#Y=20》
    《图片 #宽=75，#高=83，#源=“./res/img/鸡.jpg”，#适应=“覆盖”，#X=180，#Y=20》
    《图片 #宽=75，#高=83，#源=“./res/img/鸡.jpg”，#适应=“覆盖”，#X=265，#Y=20》
    
    《图片 #名=“#篮子”，#宽=80，#高=73，#源=“./res/img/篮子.jpg”，#适应=“覆盖”，#X=140，#Y=530》
    
    《文本标签 #名=“#得分标签”，#X=15，#Y=630，#Z=9，#字号=20，#文本=“0”》
    
    《按钮 #名=“#重新开始按钮”，#X=150，#Y=620，#Z=9，#字号=15，#文本=“重新开始”，#可见=“隐藏”》
】》

#运行中 =  真
#界面更新中  = 假
#下一按键 = “”
#蛋列表 = 【】
#得分 = 0

@最大蛋数 = @【】｛
    如果 #得分 小于 2 ｛
        返回 1
    ｝再则 #得分 小于 6 ｛
        返回 2
    ｝再则 #得分 小于 15 ｛
        返回 3
    ｝再则 #得分 小于 22 ｛
        返回 4
    ｝否则 ｛
        返回 5
    ｝
｝

@最小蛋Y = @【】｛
    #最小：999
    遍历 #蛋列表 为 #蛋｛
        如果 #蛋#Y 小于 #最小｛
            #最小 = #蛋#Y
        ｝
    ｝
    返回 #最小
｝

#窗@键盘按下命令 = @【#按键】｛
    #下一按键 = #按键
｝

@添加蛋 = @【】｛
    #序号：#数学库@随机数、4
    #X = 37 + 85 * #序号
    #蛋： @生成图形、《图片 #宽=28，#高=39，#源=“./res/img/蛋.jpg”，#适应=“覆盖”，#X=#X，#Y=105》
    #窗@加子、#蛋
    @追加【#蛋列表，#蛋】
｝

@更新蛋 = @【】｛
    #篮X ： #篮子#X

    当 #序：0，#序 小于 @取长度【#蛋列表】，#序 自加 1｛
        #蛋：#蛋列表#序
        #蛋X：#蛋#X
        #蛋Y：#蛋#Y
        
        如果 #蛋Y 大于 520 且 #蛋X 大于 #篮X 且 #蛋X 小于 #篮X + 62｛
            #窗@减子、#蛋
            @取出【#蛋列表，#序】
            #序 自减 1
            #得分 自加 1
            #得分标签#文本 = #得分 + “”
            继续
        ｝
    
        如果 #蛋Y 小于 570｛
            #蛋#Y = #蛋Y + 4
        ｝否则 ｛
            #运行中 = 假
            #蛋#源 = “./res/img/碎蛋.png”
            #蛋#宽 = 42
            #蛋#高 = 34
            #重新开始按钮#可见 = “可见”
        ｝
    ｝
｝

#定时器 = @定时任务【30，｛
    如果 取反 #运行中 ｛ 返回  空 ｝
    如果 #界面更新中 ｛ 返回  空 ｝
    #界面更新中 = 真

    如果 #下一按键 等于 “Left” 且 #篮子#X 大于 6｛
        #篮子#X  -= 6
    ｝
    如果 #下一按键 等于 “Right” 且 #篮子#X 小于 265｛
        #篮子#X += 6
    ｝
    
    @更新蛋【】
    
    如果 @取长度【#蛋列表】 小于 @最大蛋数【】 ｛
        #某：#数学库@随机数、@最大蛋数【】
        如果 #某+1 大于等于 @最大蛋数【】 且 @最小蛋Y【】 大于 250｛
            @添加蛋【】
        ｝
    ｝
    
    #界面更新中 = 假
    
｝】

#重新开始按钮@命令 = @【】｛
    #得分 = 0
    #得分标签#文本 = “0”
    遍历 #蛋列表 为 #蛋｛
        #窗@减子、#蛋
    ｝
    #蛋列表 = 【】
    #运行中 = 真
    #重新开始按钮#可见 = “隐藏”
｝



@显示窗口、#窗
@启动定时任务、#定时器




